1973

BIѲSTAT, A Computer Program Providing Simple Statistics For Biological Samples

J. M. Hillerud
Midland Lutheran College

BIΘSTAT, A COMPUTER PROGRAM PROVIDING SIMPLE STATISTICS FOR BIOLOGICAL SAMPLES

J. M. Hillerud
Department of Earth and Space Science,
Midland Lutheran College, Fremont, Nebraska 68025

ABSTRACT: A computer program developing simplė statistics (N, MIN, MAX, RANGE, MEAN, MEDN, SDEV, CVAR, and standard errors of the latter four) is available at the University of Nebraska State Museum, Lincoln.

Data are read into an array of 40 measurements (six columns; maximum 9999.9mm) for a sample maximum of 50 specimens. The "shape" of the array may be adjusted within the maximum "area" of 2000 F-field nodes in each data-set. The data-card input field provides IDENTIFICATION (cc 1-8), CARD (an A, B, C or D in c 9 indicates measurements 1-10, 11-20, 21-30 and 31-40 respectively within the array), SEX (M or F in c 10), SITE (code in cc 11-12 indicates stratigraphic or geographic location), POSITION (L or R in c 13 shows the skeletal geography), AGE-CODE (e.g., "JUV" in cc 15-17 indicates an immature specimen), BONE-CODE (consecutive numbers for all mammalian skeletal elements in cc 18-20), and ten MEASUREMENTS of cc 6 each. Verification subroutines with data-in-error messages are provided.

Output may be on cards, tape or printout. Preliminary statements (sample identification, measurement descriptions), data and statistical results are associated for each data-sequence. The program is open-ended.

In 1969-70, I took measurements on a large number of skull and postcranial elements of Holocene bison in the collections of the University of Nebraska State Museum, which had been obtained from several High Plains sites. These measurements, described and illustrated elsewhere (Hillerud, 1972b), were recorded directly on computer coding-forms and were punched as a series of card data-decks. See Hillerud (1972a, pp 39-40) for a discussion of the repeatability and reproduceability of these measurements. The computer program described below was used to generate simple univariate statistics on these data.

Although every computing center has an equivalent simple-statistics program in its files, most programs are not specifically designed to accommodate biological data. PRΘGRAM BIΘSTAT is presented here because:

(a) its range of measurements, from 0.1 to 9999.9mm, covers a major part of the macrobiological size spectrum,

(b) measurement definitions, other pertinent descriptive information and a list of the raw data are printed in proximity to the statistical printout for each data-set, facilitating its interpretation,

(c) the raw measurements can be recorded directly upon computer coding-forms, and are printed in separated columns, providing easier checking for errors in data-cards,

(d) the verification subroutine provides machine identification of mixed data-sets, helping to prevent the generation of improper statistics. If cards with errors are encountered, the subroutine stores the contents of the cards for later printout as DATA IN ERROR messages. But it allows the operator to control grouping of selected data-sets, in order to obtain a larger sample, by defining a few control-cards as "comment-cards." The verification of raw data is almost a necessary element in the generation of biostatistics, but it is not commonly provided in standard programs.

(e) the program is economical of CPU time and is open-ended. It can provide output on cards to be used as data-decks for more sophisticated statistical programs.

Initially, the program was called "PRΘGRAM SIMPSTAT" (Hillerud, 1970b, pp. 50-56, Table 3, 1-12; Hillerud, 1970a, pp. 8-9) but when we recognized that the data-array (the "z-array") generated by the first three subroutines could be used by computational subroutines other than simple univariate statistics, the program was given its present title. A new subroutine, INDEX, which divides each measurement from a single specimen by every other measurement, is being developed.

I am obliged to C. R. Montgomerie, Northern Alberta Institute of Technology, Edmonton, Canada, for initial discussions in 1968 which led to the first draft of "Program Simpstat." J. D. Smith, University of Nebraska Computing Center, debugged and amended the first draft, and wrote the verification subroutine to my specifications and complete satisfaction in 1970. J. M. Inguanzo, of the same department, aided in editing the program for publication in 1972.

I am grateful to the University of Nebraska State Museum, which provided funds for computer consultation and CPU time to develop the program in 1970 as a part of my graduate research, and to The Society of the Sigma Xi, which awarded me a Grant-in-Aid of research of $200.00 to partially defray publishing costs.

Drs. C. B. Schultz and H. L. Gunderson, University of Nebraska State Museum, and D. F. Costello, University of Nebraska Computing Center, critically read this report. I am indebted to these gentlemen for their interest and editorial criticism, but I accept responsibility for all errors remaining.

Characteristics of the Program:

The computer PRΘGRAM BIΘSTAT utilizes raw data coded in the format described below (see also Figure la), placing the numerical data in an array of 2000 F-field nodes, initially 40 measurements by 50 specimens per sample, accepting the data as follows:

cc 1-8 : specimen identification number,

c 9 : card-identification code: the letters refer to datum-positions; "A" to measurements 1-10, "B" to 11-20, "C" to 21-30 and "D" to 31-40,

c 10 : sex: "M" refers to male, "F" to female, "Blank" is not determined,

cc 11-12: site: a letter-code for geographic or stratigraphic location,

c 13 : position: skeletal geography; "L" refers to left, "R" to right, "Blank" is not applicable,

cc 15-17: age: a code for tooth-replacement or tooth-wear; for postcranial elements, "JUV" indicates juvenile specimens, "Blank" is mature,

cc 18-20: bone-code: refers the numerical data to a specific element, as listed in Table 1. Verification of this column reduces the possibility of mixed-element statistics being generated,

cc 21-80: (ten) measurements (per card), written in an F6.2 format (xxxx.x).

The program recognizes data-sets by the initial code-word "DATA" in cc 1-4. The end of an individual data-deck is identified by four "Y's" (YYYY) in cc 1-4. The conclusion of the complete data-deck is identified by eight "Y's" (YYYYYYYY) in cc 1-8.

The main program "calls" five subroutines sequentially:

(1) GETPREL accepts and prints explanatory data of any length in an alphanumeric field. These data are immediately available as printout to the researcher. The bone-code (Table 1) should be included in this section.

(2) GETDAT accepts and prints numerical data-sets presented for computation, and assigns the individual measurements to the computation (Z-) array. If an improper datum is encountered, this subroutine stores the contents of the card for later printout as a DATA IN ERROR message at the end of the run.

(3) VERIFY checks card order and homogeneity of the data-set. For example, if a B-card ("B" in c 9) with data for array-spaces 11-20 is encountered without a correct preceding A-card (with data for spaces 1-10), all information on the B-card is stored and later printed as a DATA IN ERROR message. The subroutine also verifies the data-sets by sex, site, position, age and bone-code. Sex, site, position and age-categories are removed from the program as it is reproduced here, allowing mixed data-sets of sufficient size for valid statistical analysis to be run. These control-cards remain in the subroutine deck as comment-cards, and can be re-activated by duplicating them without the "C***" in cc 1-4.

(4) PRØCES generates statistics upon the z-array data-sets in several sequences. The formulae used in this subroutine are those presented in

Simpson, *et. al.* (1960, chapters 5-6 and pp. 166-167). The statistical parameters N, RMIN, RMAX and RMEDN are developed by computer counting techniques. All formulae are translated into FORTRAN-IV computer language.

(5) PUTDAT prints the statistical computations on the line-printer, and as a punched deck if desired. The printed array includes space for 40 lines of information, of which only the number of measurement parameters are used, and the rest show as "zero". Line 41 records the total number of specimens in the sample. Figure 1(b,c,d) illustrates the printout of an abbreviated data-set.

To avoid possible typographical errors, the working PRΘGRAM BIΘSTAT is presented in facsimile as Table 2.

As here presented, PRΘGRAM BIΘSTAT accepts data in an array of 40 measurements by 50 specimens per data-set sample. A worker wishing to process fewer measurements on a larger sample can change the "shape" of the array by rewriting only a few statements in the program. For example, the following changes will produce a revised program which will accept 20 measurements on 100 specimens. The lines to be changed are marked with marginal dots in the program listing, Table 2.

In JCL machine control, rewrite the job name "BIΘSTAT40" as "BIΘSTAT20", to define the change.

In the main program, line 4, CΘMMΘN line 3, change "Z(51,40)" to "Z(101,20)". Duplicate six cards, and insert one for the main program CΘMMΘN statement, and one for the CΘMMΘN statements in each of the five subroutines.

In the main program, line 9, change "N40=40" to "N40=20", and line 10, change "N50=50" to "N50=100".

No further changes are necessary. Other adjustments to the "shape" of the array can be easily developed, by changing the nine cards listed above.

Figure 1

Coding form with data prepared for PRΘGRAM BIΘSTAT (a). The printout shows measurement definitions (b), listing of measurement data (c) and statistics printout (d) on a data-set of *Bison* radius-ulnae precessed by BIΘSTAT.

a

MUID	CARD SEX SITE POSID	AGE	BONE	MEAS (10) → 1	2	3	4	5	6
3052-65A	MLL		66	459.0	429.0	344.0	80.0	131.0	97.0
3053-65A	MLL		66	411.0	393.0	316.0	65.0	100.0	86.0
3054-65A	MLL		66	416.0	399.0	321.0	66.0	106.0	83.0
3056-65A	MLL		66	—	—	315.0	—	—	85.0
3058-65A	MLL		66	408.0	398.0	320.0	68.0	109.0	89.0
3060-65A	MLL		66	430.0	419.0	319.0	74.0	132.0	99.0
				11	12	13			
3052-65B	MLL		66	32.0	90.0	66.0			
3053-65B	MLL		66	21.0	81.0	67.0			
3054-65B	MLL		66	21.0	75.0	57.0			
3056-65B	MLL		66	—	75.0	55.0			

b

```
PRELIMINARY STATEMENT  - RADIUS-ULNA -BONE CODE NO. 66.
 1        TOTAL LENGTH OF ELEMENT.
 2        LENGTH OF ULNA.
 3        LENGTH OF RADIUS.
 4        ANTERIOR-POSTERIOR WIDTH OF OLECRANON.
 5        LENGTH OF OLECRANON TO SEMILUNAR NOTCH.
 6        LATERAL WIDTH OF PROXIMAL EXTREMITY.
 7        WIDTH OF PROXIMAL ARTICULATING SURFACE, RADIUS.
 8        ANTERIOR-POSTERIOR WIDTH OF RADIUS, PROXIMAL EXTREMITY.
 9        LATERAL WIDTH OF RADIUS SHAFT AT NARROWEST POINT.
10        ANTERIOR-POSTERIOR WIDTH OF RADIUS SHAFT AT SAME POINT.
11        WIDTH (ANT-POST) OF ULNA AT POINT OF SHAFT FUSION.
12        WIDTH OF RADIUS-ULNA, DISTAL EXTREMITY.
13        ANTERIOR-POSTERIOR WIDTH, DISTAL EXTREMITY OF RADIUS-ULNA.
YYYY
```

c

DATA MIXED DECK: ADULT AND JUVENILE RADIUS-ULNAE, MILBURN (9)

3052-65A	L	66	459.00	429.00	344.00	80.00	131.00	97.00	91.00	52.00	55.00
3052-65B	L	66	32.00	90.00	66.00	0.00	0.00	0.00	0.00	0.00	0.00
3053-65A	L	66	411.00	393.00	316.00	65.00	100.00	86.00	83.00	46.00	47.00
3053-65B	L	66	21.00	81.00	67.00	0.00	0.00	0.00	0.00	0.00	0.00
3054-65A	L	66	416.00	399.00	321.00	66.00	106.00	83.00	80.00	45.00	41.00
3054-65B	L	66	21.00	75.00	57.00	0.00	0.00	0.00	0.00	0.00	0.00
3056-65A	L	66	0.00	0.00	315.00	0.00	0.00	85.00	83.00	47.00	43.00
3056-65B	L	66	0.00	75.00	55.00	0.00	0.00	0.00	0.00	0.00	0.00
3058-65A	L	66	408.00	398.00	320.00	68.00	109.00	89.00	86.00	46.00	46.00
3058-65B	L	66	23.00	81.00	56.00	0.00	0.00	0.00	0.00	0.00	0.00
3060-65A	L	66	430.00	419.00	319.00	74.00	132.00	99.00	92.00	[illegible]	
3060-65B	L	66	30.00	86.00	69.00	0.00	0.00	[illegible]			(10) ↗
3064-65A	L	66	478.00	458.00	364.00	81.[illegible]					
3064-65B	L	66	29.00	91.00	[illegible]						
3071-65A	L	66	[illegible]								
3071-65B	'										

d

SECVAR ↘

ID	MEAS	N	MIN	(MEAN)	MAX	RANGE	MEDIAN	S DEV	CVAR	SEMEAN	SEMEDN	SESDEV	SECVA
66	1	31	380.00	(422.61)	493.00	113.00	413.00	29.76	7.04	5.35	6.68	3.78	0.8
66	2	31	370.00	(405.58)	470.00	100.00	399.00	26.80	6.61	4.81	6.02	3.40	0.8
66	3	40	296.00	(323.77)	375.00	79.00	320.00	17.97	5.55	2.84	3.55	2.01	0.6
66	4	33	56.00	(69.00)	82.00	26.00	66.00	6.91	10.02	1.20	1.50	0.85	1.2
66	5	34	91.00	(112.15)	147.00	56.00	107.00	15.70	14.00	2.69	3.37	1.90	1.7
66	6	39	78.00	(88.62)	112.00	34.00	86.00	8.70	9.81	1.39	1.74	0.98	1.1
66	7	39	75.00	(84.05)	101.00	26.00	82.00	7.19	8.55	1.15	1.44	0.81	0.9
66	8	39	42.00	(46.85)	60.00	18.00	46.00	4.25	9.08	0.68	0.85	0.48	1.[illegible]
66	9	40	39.00	(46.05)	61.00	22.00	44.00	6.07	13.17	0.96	1.20	0.68	1.[illegible]
66	10	40	24.00	(29.40)	41.00	17.00	28.00	3.81	12.97	0.60	0.75	0.43	1.[illegible]
66	11	33	19.00	(24.00)	34.00	15.00	22.00	4.18	17.41	0.73	0.91	0.51	2.[illegible]
66	12	40	65.00	(79.00)	99.00	34.00	76.00	8.08	10.23	1.28	1.60	0.90	1.[illegible]
66	13	39	52.00	(65.00)	88.00	36.00	62.00	9.19	14.14	1.47	1.84	1.04	1.[illegible]
66	14	0	0.00	(0.00)	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.
66	15	0	0.00	(0.00)	0.00	0.00	0.00	0.00	0.00	0.00	0.00	[illegible]	
66	16	0	0.00	(0.00)	0.00	0.00	0.00	0.00	0.00	[illegible]	[illegible]	0.00	0
66	17	0	0.00	(0.00)	0.00	0.00	0.00	[illegible]	[illegible]	0.00	0.00	0.00	0
66	18	0	0.00	(0.00)	0.00	[illegible]	[illegible]	0.00	0.00	0.00	0.00	0.00	0
66	19	0	0.0[illegible]	[illegible]	[illegible]	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0
[illegible]	[illegible]		[illegible]	0.00)	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0
66	38	0	0.00	(0.00)	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0
66	39	0	0.00	(0.00)	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0
66	40	0	0.00	(0.00)	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0

THE NUMBER OF SPECIMENS IS 43

Table 1. The mammalian bone-code, listing the elements of the mammalian skeleton with identifying code-numbers assigned to each element. The number is placed in cc 18-20 on each data card. Note the aberrant 900-numbers for unidentified teeth; these illustrate a technique for revision of the code, presented here as a "third approximation."

```
C
C
        CRANIAL ELEMENTS
C
      1 SKULL FRAGMENT 8   UC-1          16  LI-1          24  LP-3
      2 MANDIBLE RAMUS 9   UP-1          17  LI-2          25  LP-4

        TEETH         10   UP-2          18  LI-3         925 UNIDENT. LPM
      3  UI-1         11   UP-3          19  LI-4          26  LM-1
      4  UI-2         12   UP-4          20  LI-5          28  LM-2
      5  UI-3        912 UNIDENT. UP-   920 UNIDENT. LI-   28  LM-3
      6  UI-4         13   UM-1          21  LC-1         928 UNIDENT. LM-
      7  UI-5         14   UM-2          22  LP-1          29 INIDENT. TOOTH
    907 UNIDENT. UI-  15   UM-3          23  LP-2          30 GREAT CORNU HYOID
                     915 UNIDENT. UM-                      31 MINOR CORNU HYOID
C
        AXIAL ELEMENTS
C
     32  CV-1 ATLAS    40  TV-1          48  TV- 9         56  LV-3
     33  CV-2 AXIS     41  TV-2          49  TV-10         57  LV-4
     34  CV-3          42  TV-3          50  TV-11         58  LV-5
     35  CV-4          43  TV-4          51  TV-12         59  LV-6
     36  CV-5          44  TV-5          52  TV-13         60 UNIDENT LV
     37  CV-6          45  TV-6          53 UNIDENT TV     61 SACRUM
     38  CV-7          46  TV-7          54  LV- 1         62 CAUDAL V. UNDIFF.
     39 UNIDENT CV     47  TV-8          55  LV- 2         63 ----------
C
        FORE-LIMB ELEMENTS
C
     64 SCAPULA        70 ULNAR C.       76  MC-2
     65 HUMERUS        71 ACCESSORY C.   77  MC-3 (CANNON)
     66 RADIUS(-ULNA)  72 C(2+3) C.      78  MC-4
     67 ULNA           73 C-4 C.         79  MC-5
     68 RADIAL C.      74 CARPAL UNDIF.  80 UNIDENT MC  (SPLINT)
     69 INTERMEDIATE C.75 METACARPAL-1   81  ----------
C
        PHALANGES          ( PHALANX ONE, DIGIT TWO)
C
        PHALANGES          ( PHALANX ONE, DIGIT TWO)
C
     82  P-I-D-1       87  P-II-D-1      92  P-III-D-1     97 UNIDENT PHALANX
     83  P-I-D-2       88  P-II-D-2      93  P-III-C-2
     84  P-I-D-3       89  P-II-D-3      94  P-III-D-3     NOTE. FOR ARTIODAC-
     85  P-I-D-4       90  P-II-D-4      95  P-III-C-4      TYLS, -RIGHT- PHAL-
     86  P-I-D-5       91  P-II-D-5      96  P-III-C-5      ANGES ARE DIGIT -3-
                                                            AND -LEFT- PHALANGES
                                                            ARE DIGIT -4-.
        HIND-LIMB ELEMENTS
C
     97 PELVIS        103 LAT.MAL.FIB. 109  MT-2
     98 FEMUR         104 T-(1)        110  MT-3 (CANNON)
     99 TIBIA         105 T-(2+3)      111  MT-4
    100 FIBULA        106 T-(C+4)      112  MT-5
    101 CALCANEUM     107 UNIDENT TARS.113 UNIDENT MT (INCLUDES SPLINT).
    102 ASTRAGALUS    108  MT-1        114 ----------
C
        RIB-CAGE ELEMENTS
C
    115  R- 1         119  R- 5        123  R- 9         127  R-13
    116  R- 2         120  R- 6        124  R-10         128 UNIDENT. RIB
    117  R- 3         121  R- 7        125  R-11         129 COSTAL CART.
    118  R- 4         122  R- 8        126  R-12         130 ---------
C
        ACCESSORY ELEMENTS
C
    131 CLAVICLE      134 XIPHISTERNUM 137 SESAMOID DIST140  ----------
    132 STERNEBRA     135 PATELLA      138 BACULUM       141  ----------
    133 MANUBRIUM     136 SESAMOID PROX139 OS CLITORIS   142 --------
```

Table 2. PRΘGRAM BIΘSTAT, in facsimile.

2-1

JCL Machine Control

```
//HILLERUD JOB MUSEAA304300,'BIOSTAT40',MSGLEVEL=1,PRTY=7,CLASS=A          JOB 415
//WATFIV EXEC WATFIV,REGION.GO=200K
XXWATFIV  PROC    PROG=WATFIV,PLOT='DUMMY                                   00001000
***  IF YOU WISH TO PLOT SPECIFY PLOT='VOL=PRIVATE' ON THE EXEC CARD         00002000
***  AND PROG=WATPLOT                                                        00003000
XXGO     EXEC    PGM=&PROG,REGION=140K                                       00004000
IEF653I SUBSTITUTION JCL - PGM=WATFIV,REGION=140K
//SYSPRINT DD SYSOUT=(G,,0200)
X/SYSPRINT  DD  SYSOUT=A                                                     00005000
XXWATLIB DD  DSN=SYS1.WATLIB,UNIT=2314,VOL=SER=SYSRS1,DISP=SHR,              00006000
XX      DCB=(LRECL=80,BLKSIZE=960,RECFM=FB)                                  00007000
XXFT01F001 DD DUMMY                                                          00008000
XXFT02F001 DD DUMMY                                                          00009000
XXFT03F001 DD DUMMY                                                          00010000
XXFT04F001 DD DUMMY                                                          00011000
XXFT05F001  DD  DDNAME=SYSIN                                                 00012000
//GO.FT06F001 DD SYSOUT=(G,,0200),DCB=(RECFM=FBA,LRECL=133,
//    BLKSIZE=133)
X/FT06F001  DD  SYSOUT=A,DCB=(RECFM=FBA,LRECL=133,BLKSIZE=133)               00013000
XXFT07F001  DD  SYSOUT=B                                                     00014000
XXPLOTTAPE  DD  &PLOT,UNIT=TAPE,DISP=NEW,LABEL=(,NL)                         00015000
IEF653I SUBSTITUTION JCL - DUMMY ,UNIT=TAPE,DISP=NEW,LABEL=(,NL)
//GO.SYSIN DD *
//
IEF142I - STEP WAS EXECUTED - COND CODE 0000
IEF373I STEP /GO      / START 72252.1452
IEF374I STEP /GO      / STOP  72252.1453 CPU    0MIN 06.79SEC MAIN 200K LCS    0K
                                                                  extended line

STEP   /RGH/RGL/CPU TIME/DP/   DISK/   TAPE/   2250/   1403/   CELL/  2540R/  2540P/
GO       200   0    6.79  7      12      0       0       0       0       0       0
                            continued  SYSIN/H-PRINT/H-PUNCH/EFF-COR-TM/   CHARGE
IEF375I  JOB /HILLERUD/ START 72252.1452    526     606      0       7.51     0.89
IEF376I  JOB /HILLERUD/ STOP  72252.1453 CPU   0MIN 06.79SEC

      JOBNAME     ACCOUNT           PRTY   CLASS    TOTAL CHARGE          BALANCE
      HILLERUD  MUSEAA304300         07      A           0.98               49.81
```

Program

```
$JOB
C      PROGRAM BIOSTAT
C      THIS PROGRAM DEVELOPES SIMPLE STATISTICS ON BIOMETRIC DATA.  DATA-
C      INPUT CARDS HAVE THE FOLLOWING FORMAT.
C          CC- 1 TO  8 -- MUSEUM IDENTIFICATION NUMBER
C          C -       9 -- CARD IDENTIFICATION (A,B,C,D), WITH MEASUREMENTS 01-10
C                          ON THE A-CARD, M--11-20 ON THE B-CARD, M--21-30 ON THE
C                          C-CARD, AND M--31-40 ON THE D-CARD.
C          C -      10 -- SEX (M,F, ): MALE, FEMALE, NOT DETERMINED.
C          CC-11 TO 12 -- SITE:  (RESERVED FOR STRATIGRAPHIC HORIZON).
C          C -      13 -- POSITION (L,R, ): LEFT, RIGHT, NOT APPLICABLE.
C          CC-15 TO 17 -- AGE-CODE (VARIOUS, EG., JUV): BLANK INDICATES MATURE.
C          CC-18 TO 20 -- BONE-CODE  (INTEGER). ELEMENT IDENTIFICATION. SEE
C                          LISTING WHICH FOLLOWS.
C          CC-21 TO 80 -- MEASUREMENTS (10F6.2). MEASUREMENT RANGE 0.1 TO 9999.9
C                          MILLIMETERS.
C
       INTEGER*4 CARD,SEX,SITE,POSID,AGE,BONE
       REAL*4    MEASUR
       REAL*8    MUID,HMUID
       COMMON MUID(201), MEASUR(201,10), RMIN(40), RMEAN(40), RMAX(40),
     1        RRANGE(40), RMEDN(40), STNDEV(40), RCVAR(40), SEMEAN(40),
     2        SESDEV(40), SECVAR(40), SEMEDN(40), CARD(201), SEX(201),
     3        SITE(201), POSID(201), AGE(201), BONE(201), Z(51,40),
     4        ICNTA, DONE, ILPP, IN, IO, IE, SUM(40), N10, N40, N50,
     5        N51, N55, N200, N201, I
       IN=5
       IO=6
       IE=8
C        TO TAILOR THIS PROGRAM TO A SPECIFIC MACHINE, CHANGE THE INPUT AND
C        OUTPUT CODES IN THE LINES ABOVE TO CONFORM WITH THOSE USED BY THE
C        NEW MACHINE.  "IN" IS THE READ-IN CODE, "IO" IS THE READOUT CODE, IE
C        IS A READOUT CODE WHICH CAN BE USED TO READ OUT DATA IN ERROR, BY
C        ACTIVATING CERTAIN CONTROL CARDS MARKED "C***" IN CC 1-4.
```

2-2

```
      N10=10
      N40=40
      N50=50
      N51=N50+1
      N55=55
      N200=200
      N201=N200+1
      DONE=0.0
    1 CALL GTPREL
      IF(DONE.EQ.2.0)GO TO 5
      CALL GETDAT
      CALL VERIFY
      CALL PROCES
      CALL PUTDAT
      IF(DONE.EQ.0.0)GO TO 1
      WRITE(IO,4)
    4 FORMAT('1')
    5 STOP
      END

      SUBROUTINE GTPREL
      DIMENSION LINE(20)
      INTEGER*4 DATAWD
      DATA DATAWD,WYEEE4/'DATA','YYYY'/
      INTEGER*4 CARD,SEX,SITE,POSID,AGE,BONE
      REAL*4    MEASUR
      REAL*8    MUID,HMUID

      COMMON MUID(201), MEASUR(201,10), RMIN(40), RMEAN(40), RMAX(40),
     1       RRANGE(40), RMEDN(40), STNDEV(40), RCVAR(40), SEMEAN(40),
     2       SESDEV(40), SECVAR(40), SEMEDN(40), CARD(201), SEX(201),
     3       SITE(201), POSID(201), AGE(201), BONE(201), Z(51,40),
     4       ICNTA, DONE, ILPP, IN, IO, IE, SUM(40), N10, N40, N50,
     5       N51, N55, N200, N201, I
    1 LINECT=0
      WRITE(IO,2)
      ILPP=0
    2 FORMAT('1'////)
    5 READ(IN,10,END=30)LINE
   10 FORMAT(20A4)
      IF(LINE(1).EQ.DATAWD) GO TO 50
      WRITE(IO,20)LINE
      LINECT=LINECT+1
      IF(LINECT.GT.N40)GO TO 1
   20 FORMAT(1H ,18X,20A4)
      ILPP=ILPP+1
      IF(ILPP.LE.N55)GO TO 5
      WRITE(IO,25)
   25 FORMAT('1')
      ILPP=0
      GO TO 5
   30 DONE=2.0
   50 WRITE(IO,60)LINE
   60 FORMAT('1',////,18X,20A4)
      RETURN
      END

      SUBROUTINE GETDAT
      REAL*8 WYEEE4,WYEEE8
      INTEGER*4 ACRD
      DATA WYEEE4,WYEEE8,ACRD/'YYYY    ','YYYYYYYY','A'/
      INTEGER*4 CARD,SEX,SITE,POSID,AGE,BONE
      REAL*4    MEASUR
      REAL*8    MUID,HMUID
      COMMON MUID(201), MEASUR(201,10), RMIN(40), RMEAN(40), RMAX(40),
     1       RRANGE(40), RMEDN(40), STNDEV(40), RCVAR(40), SEMEAN(40),
     2       SESDEV(40), SECVAR(40), SEMEDN(40), CARD(201), SEX(201),
     3       SITE(201), POSID(201), AGE(201), BONE(201), Z(51,40),
     4       ICNTA, DONE, ILPP, IN, IO, IE, SUM(40), N10, N40, N50,
     5       N51, N55, N200, N201, I
  110 FORMAT('1'///////)
      ICNTA=0
C        ICNTA = THE NUMBER OF SPECIMENS...
      DO 2 IZ=1,N201
      CARD(IZ)=0
      SEX(IZ)=0
      SITE(IZ)=0
      POSID(IZ)=0
      AGE(IZ)=0
      BONE(IZ)=0
      DO 1 IY =1,N10
    1 MEASUR(IZ,IY)=0.0
    2 CONTINUE
```

2–3

```
          DO 3 IZ=1,N51
          DO 3 IX=1,N40
        3 Z(IZ,IX)=0.0
     C    READ DATA CARDS (MAXIMUM OF 200 PER DATA SET)
     C    I IS THE NUMBER OF CARDS READ IN.
          I=0
        5 I=I+1
          READ(IN,10,END=99)MUID(I),CARD(I),SEX(I),SITE(I),POSID(I),AGE(I),
         1BONE(I),(MEASUR(I,J),J=1,N10)
       10 FORMAT(A8,2A1,A2,A1,1X,2A3,10F6.2)
          IF(MUID(I).EQ.WYEEE8)GO TO 99
          IF(I.EQ.N201)GO TO 80
          IF(MUID(I).EQ.WYEEE4)GO TO 100
          WRITE(IO,70)MUID(I),CARD(I),SEX(I),SITE(I),POSID(I),AGE(I),
         1BONE(I),(MEASUR(I,J),J=1,N10)
       70 FORMAT(' ',18X,A8,2A1,A2,A1,1X,2A3,1X,    10(F6.2,1X) )
          ILPP=ILPP+1
          IF(ILPP.LE.N55)GO TO 75
          ILPP=0
          WRITE(IO,73)
       73 FORMAT ('1')
       75 IF(CARD(I).EQ.ACRD)ICNTA=ICNTA+1
          IF(ICNTA.GT.N50)GO TO 85
          GO TO 5
       80 IF(MUID(I).EQ.WYEEE8)GO TO 99
          IF(MUID(I).EQ.WYEEE4)GO TO 100
     C    MORE THAN 50 A-CARDS OR MORE THAN 200 DATA-CARDS HAVE BEEN DETECTED
     C    WITHOUT AN END-OF-SAMPLE YYYY-CARD.  PROCESSING OF THE FIRST 50 DATA-SET
     C    WILL CONTINUE.
       85 CONTINUE
          IP=I
          WRITE(IO,90)MUID(IP),CARD(IP),SEX(IP),SITE(IP),POSID(IP),
         1AGE(IP),BONE(IP),(MEASUR(IP,J),J=1,N10)
       90 FORMAT(' TRUNCATED DATA',2X,A8,2A1,A2,A1,1X,2A3,10F6.2)
          IF(I.GT.N200)I=N200
          GO TO 5
       99 DONE=1.0
      100 I=I-1
          RETURN
          END
     C       THE FORMAT STATEMENT 110 CONTROLS MARGIN SPACING.  IT IS REFERENCED
     C       LINE 1, SUBROUTINE GETDAT.
WARNING**  FORMAT STATEMENT    110 IS UNREFFRENCED

          SUBROUTINE VERIFY
     C    THIS SUBROUTINE CHECKS FOR SAMENESS IN THE SEVEN IDENTIFICATION
     C    FIELDS PER CARD GROUP- AN A,B,C,D IN COL 9, AN M,F IN COL 10,
     C    SITE NUMBER IN COL 11-12, AN L,R IN COL 13, A JUV IN COL 15-17,
     C    AND PUTS THE A,B,C,D MEASUREMENTS IN THE Z-ARRAY.
          INTEGER*4 ABCD(4), HCARD, HSEX, HPOSID, HSITE, HAGE, HBONE
          DATA ABCD/'A','B','C','D'/
          INTEGER*4 CARD,SEX,SITE,POSID,AGE,BONE
          REAL*4     MEASUR
          REAL*8     MUID,HMUID
          COMMON MUID(201), MEASUR(201,10), RMIN(40), RMEAN(40), RMAX(40),
         1       RRANGE(40), RMEDN(40), STNDEV(40), RCVAR(40), SEMEAN(40),
         2       SESDEV(40), SECVAR(40), SEMEDN(40), CARD(201), SEX(201),
         3       SITE(201), POSID(201), AGE(201), BONE(201), Z(51,40),
         4       ICNTA, DONE, ILPP, IN, IO, IE, SUM(40), N10, N40, N50,
         5       N51, N55, N200, N201, I
          M=0
          J=0
          L=0
        4 J=J+1
          IF(J.GT.I)GO TO 250
        5 IF(CARD(J).NE.ABCD(1))GO TO 200
          I1=0
          I3=1
          HMUID=MUID(J)
          HCARD=CARD(J)
     C*** HSEX=SEX(J)
     C*** HSITE=SITE(J)
     C*** HPOSID=POSID(J)
     C*** HAGE=AGE(J)
     C    TO ACTIVATE VERIFICATION OF SAMENESS IN THE IDENTIFICATION FIELDS SITE,
     C    SEX, POSID, AGE  ABOVE, DUPLICATE THE CARDS WITH ASTERISKS AS VALID
     C    STATEMENT CARDS, NOT COMMENT CARDS.
     C    AS WRITTEN, THE SUBROUTINE ALLOWS GROUPED-SAMPLE STATISTICS TO BE
     C    COMPUTED.
          HBONE=BONE(J)
          GO TO 30
```

```
   20 J=J+1
      IF(J.GT.I)GO TO 250
      IF(CARD(J).NE.ABCD(I3))GO TO 5
      IF(MUID(J).NE.HMUID)GO TO 200
C***  IF(SEX(J).NE.HSEX)GO TO 200
C***  IF(SITE(J).NE.HSITE)GO TO 200
C***  IF(POSID(J).NE.HPOSID)GO TO 200
C***  IF(AGE(J).NE.HAGE)GO TO 200
C      SEE PREVIOUS COMMENT STATEMENT.
      IF(BONE(J).NE.HBONE)GO TO 200
   30 DO 50 L=1,N10
      I1=I1+1
      IF(MEASUR(J,L).EQ.0.0)GO TO 50
      Z(N51,I1)=Z(N51,I1)+1
      IZ=Z(N51,I1)
      Z(IZ,I1)=MEASUR(J,L)
   50 CONTINUE
      I3=I3+1
      IF(I1.GE.N40)GO TO 4
      GO TO 20
  200 CONTINUE
C     THE VERIFY SUBROUTINE HAS ENCOUNTERED ERRONEOUS DATA.
C     WRITE ERROR CARD(S) AND GET NEXT A-CARD.
      WRITE(IO,220)MUID(J),CARD(J),SEX(J),SITE(J),POSID(J),AGE(J),
     1BONE(J),(MEASUR(J,K),K=1,N10)
  220 FORMAT(' DATA IN ERROR',2X,A8,2A1,A2,A1,2A3,10F6.2)
      J=J+1
      IF(J.LE.I)GO TO 5
  250 RETURN
      END

      SUBROUTINE PROCES
      INTEGER*4 CARD,SEX,SITE,POSID,AGE,BONE
      REAL*4    MEASUR
      REAL*8    MUID,HMUID
      COMMON MUID(201), MEASUR(201,10), RMIN(40), RMEAN(40), RMAX(40),
     1       RRANGE(40), RMEDN(40), STNDEV(40), RCVAR(40), SEMEAN(40),
     2       SESDEV(40), SECVAR(40), SEMEDN(40), CARD(201), SEX(201),
     3       SITE(201), POSID(201), AGE(201), BONE(201), Z(51,40),
     4       ICNTA, DONE, ILPP, IN, IO, IE, SUM(40), N10, N40, N50,
     5       N51, N55, N200, N201, I
C     CALCULATE MIN, MAX, MEAN, AND MEDIAN, RANGE,MEASUR,I,Z
C***  WRITE (IE,2)
C***2 FORMAT('1  THE Z ARRAY')
C***  DO 3 ID=1,N51
C***3 WRITE(IE,5)(Z(ID,IG),IG=1,N40)
C***5 FORMAT(' ',4(10(F6.2,4X)/5X))
C     DUPLICATION OF CARDS WITH ASTERISKS AS VALID STATEMENT CARDS, NOT
C     COMMENT CARDS, WILL WRITE THE Z-ARRAY (ALL DATA IN THE FIELD OF
C     40 MEASUREMENTS ON EACH OF 50 SPECIMENS). THIS LOOP WAS USEFUL IN
C     CHECKING TEST-DATA DURING INITIAL PRODUCTION OF PROGRAM BIOSTAT.
C***  WRITE(IE,6)J
C***6 FORMAT(' ', 5H J = ,I4)
      II=I
      DO 50 J=1,N40
      RMIN(J)=0.0
      RMAX(J)=0.0
      SUM(J)=0.0
      RMEAN(J)=0.0
      RMEDN(J)=0.0
      RRANGE(J)=0.0
      STNDEV(J)=0.0
      RCVAR(J)=0.0
      SEMEAN(J)=0.0
      SESDEV(J)=0.0
      SECVAR(J)=0.0
      SEMEDN(J)=0.0
      ITIMES=Z(N51,J)
C***  WRITE(IE,7)ITIMES
C***7 FORMAT(' ',9H ITIMES  ,I4)
      IF(ITIMES.EQ.0)GO TO 50
      IF(ITIMES.GT.N50)GO TO 50
      IHALF=(ITIMES+2)/2
      RMIN(J)=999.99
      RMAX(J)=0.0
      SUM(J)=0.0
      RMEAN(J)=0.0
      DO 9 I=1,ITIMES
      IF(Z(I,J).LT.RMIN(J))RMIN(J)=Z(I,J)
      IF(Z(I,J).GT.RMAX(J))RMAX(J)=Z(I,J)
      SUM(J)=SUM(J)+Z(I,J)
    9 CONTINUE
```

2–5

```
          DO 10 I=1,IHALF
          DO 10 K=I,ITIMES
          IF(Z(I,J).GE.Z(K,J))GO TO 10
          RHLD=Z(I,J)
          Z(I,J)=Z(K,J)
          Z(K,J)=RHLD
       10 CONTINUE
          RMEAN(J)=SUM(J)/ITIMES
          RMEDN(J)=Z(IHALF,J)
          RRANGE(J)=RMAX(J)-RMIN(J)
C***      WRITE(IE,15)RMEAN(J),RMEDN(J),RRANGE(J),SUM(J),RMAX(J),RMIN(J)
C**15 FORMAT(' ',28H MEAN MEDN RANGE SUM MAX MIN,6(F8.2,5X))
C         COMPUTE THE STANDARD DEVIATION
          RSUM=0
          DO 20 K=1,ITIMES
       20 RSUM=RSUM+(RMEAN(J)-Z(K,J))**2
          STNDEV(J)=SQRT(RSUM/ITIMES)
C         COMPUTE THE COEFFICIENT OF VARIATION
          RCVAR(J)=STNDEV(J)*100/RMEAN(J)
C***      WRITE(IE,17)STNDEV(J),RCVAR(J)
C**17 FORMAT(' ', 6HSTNDEV,2X,F8.2,2X,6HRCVAR ,1X,F8.2)
C
C         COMPUTE THE STANDARD ERRORS OF MEAN, STANDARD DEVIATION,
C         COEFFICIENT OF VARIATION AND MEDIAN
          RTIMES=ITIMES
          SEMEAN(J)=STNDEV(J)/SQRT(RTIMES)
          SESDEV(J)=STNDEV(J)/SQRT(RTIMES*2)
          SECVAR(J)=RCVAR(J)/SQRT(RTIMES*2)
          SEMEDN(J)=1.25*STNDEV(J)/SQRT(RTIMES)
C***      WRITE(IE,19)SEMEAN(J),SESDEV(J),SECVAR(J),SEMEDN(J)
C**19 FORMAT(' ',6HSEMEAN,2X,F8.2,2X,6HSESDEV,2X,F8.2,2X,
C*** 1    6HSECVAR,2X,F8.2,2X,6HSEMEDN,2X,F8.2 )
       50 CONTINUE
          I=II
          RETURN
          END
C

          SUBROUTINE PUTDAT
          INTEGER*4 CARD,SEX,SITE,POSID,AGE,BONE
          REAL*4     MEASUR
          REAL*8     MUID,HMUID
          COMMON MUID(201), MEASUR(201,10), RMIN(40), RMEAN(40), RMAX(40),
        1        RRANGE(40), RMEDN(40), STNDEV(40), RCVAR(40), SEMEAN(40),
        2        SESDEV(40), SECVAR(40), SEMEDN(40), CARD(201), SEX(201),
        3        SITE(201), POSID(201), AGE(201), BONE(201), Z(51,40),
        4        ICNTA, DONE, ILPP, IN, IO, IE, SUM(40), N10, N40, N50,
        5        N51, N55, N200, N201, I
C         WRITE TITLES FOR STATISTICAL OUTPUT
          WRITE(IO,5)
        5 FORMAT('1'////)
          WRITE(IO,10)
       10 FORMAT('I',15X,2HID,1X,4HMEAS,1X,1HN,2X,3HMIN,2X,1H(,6H MEAN ,1H),
         12X,3HMAX,3X,5HRANGE,1X,6HMEDIAN,3X,5HS DEV,2X,4HCVAR,2X,6HSEMEAN,
         21X,6HSEMEDN,1X,6HSESDEV,1X,6HSECVAR)
C         WRITE STATISTICAL VALUES
          DO 20 J=1,N40
          IZ=Z(N51,J)
          WRITE(IO,15)BONE(1),J,IZ,RMIN(J),RMEAN(J),
         1RMAX(J),RRANGE(J),RMEDN(J),STNDEV(J),
         2RCVAR(J),SEMEAN(J),SEMEDN(J),SESDEV(J),SECVAR(J)
       15 FORMAT(1H ,15X A3, 1X, I2, 1X, I3, 1X,
         1F6.2,1H(,F6.2,1H),F6.2,1X,
         28(F6.2,1X))
       20 CONTINUE
          WRITE(IO,40)ICNTA
       40 FORMAT(' ',15X,27HTHE NUMBER OF SPECIMENS IS ,I4)
          RETURN
          END

$ENTRY
```

End YYYYYYYY

```
CORE USAGE        OBJECT CODE=    17504 BYTES,ARRAY AREA=    24704 BYTES,TOTAL

                                              AREA AVAILABLE=    100448  BYTES

COMPILE TIME=      1.52 SEC,EXECUTION TIME=      4.82 SEC,  WATFIV - VERSION 1

                                                 - LEVEL 2.5 - SEPTEMBER 1971
```

REFERENCES CITED

Hillerud, J.M. 1970a. A computer program providing simple statistics for biological samples. (ABST), Proc. Nebraska Acad. Sciences, April 24-25,. Nebraska Univ. Press. pp. 34-35. (Eightieth Ann. Meeting).

Hillerud, J.M. 1970b. Subfossil High Plains Bison (volumes I-II). Ph.D. thesis, Univ. Nebraska Dept. Geol. 606p., pls. 1-9, figs. 1-9, tables 5, appendices 7.

Hillerud, J.M. 1972a. The measurement of selected bison post-cranial elements; a preliminary report. (ABST), Proc. Nebraska Acad. Sciences, April 21-22,. Nebraska Univ. Press. pp. 39-40. (Eighty-second Ann. Meeting).

Hillerud, J.M. 1972b. Comparative measurements for *Bison* skulls and selected postcranial elements.

Simpson, G.G., A. Roe and R. Lewontin. 1960. Quantitative zoology, revised edition. Harcourt, Brace and Co., New York. 440p.